// File: contracts/config/GlobalConfig.sol



// SPDX-License-Identifier: MIT
pragma solidity >=0.8.0 <0.9.0;

import "./IGlobalConfig.sol";
import "../library/Registry.sol";
import "../library/LEnumerableMetadata.sol";
import "@openzeppelin/contracts/access/AccessControlEnumerable.sol";


contract GlobalConfig is IGlobalConfig, AccessControlEnumerable {

    using LEnumerableMetadata for LEnumerableMetadata.MetadataSet;
    LEnumerableMetadata.MetadataSet private _configSet;

    // roleID -> fatherRoleID.
    // one addr in fatherRole, the addr also in children role.
    mapping(bytes32=>bytes32) public fatherRoleMap;

    constructor(address superAdmin) {
        if (superAdmin == address(0)) revert("GlobalConfig cannot be created");
        _setupRole(Registry.SUPER_ADMIN_ROLE, superAdmin);
        _configSet._init();
    }

    function supportsInterface(bytes4 interfaceId) public view virtual override(IConfig, AccessControlEnumerable) returns (bool) {
      return interfaceId == type(IGlobalConfig).interfaceId || interfaceId == type(IConfig).interfaceId || super.supportsInterface(interfaceId);
    }

    function version() external pure returns (uint256 v) {
        return 1;
    }

    function revokeRole(bytes32 role, address account) external override isRoleAdmin(role){
        require (role != Registry.SUPER_ADMIN_ROLE || getRoleMemberCount(role) > 1, "SC101: one super admin exists at a minimum");
        _revokeRole(role, account);
    }

    function renounceRole(bytes32 role, address account) external override {
        require (role != Registry.SUPER_ADMIN_ROLE || getRoleMemberCount(role) > 1, "SC101: one super admin exists at a minimum");
        super.renounceRole(role, account);
    }

    function hasRole(bytes32 role, address account) external view override(IAccessControl, IConfig) returns (bool has) {
      if(super.hasRole(role, account)){
        return true;
      }

      // the father role map default value is super admin role.
      if(super.hasRole(fatherRoleMap[role], account)){
        return true;
      }

      // if addr is role admin, so the addr has the role.
      if(super.hasRole(getRoleAdmin(role), account)){
        return true;
      }
      
      if (super.hasRole(Registry.SUPER_ADMIN_ROLE, account)) {
        return true;
      }

      return false;
    }

    function checkAdmin(address account) public view returns(bool) {
      // hasRole always check superadmin.
      return hasRole(Registry.ADMIN_ROLE, account);
    }

    modifier isAdmin() {
        require(checkAdmin(msg.sender), "SC102: invoke restricted to admin role");
        _;
    }

    modifier isSuperAdmin(){
        _checkRole(Registry.SUPER_ADMIN_ROLE, msg.sender);
        _;
    }

    modifier checkNotAdminRole(bytes32 role){
      require(role != Registry.SUPER_ADMIN_ROLE && role != Registry.ADMIN_ROLE, "SC103: ask for neither super admin nor admin role");

      _;
    }

    modifier isRoleAdmin(bytes32 role){
      require(super.hasRole(getRoleAdmin(role), msg.sender) || super.hasRole(Registry.SUPER_ADMIN_ROLE, msg.sender), "SC104: not role admin");
      _;
    }

    function setFatherRole(bytes32 role, bytes32 fatherRole) external isSuperAdmin checkNotAdminRole(role){
      require(getRoleAdmin(fatherRole) != role, "SC105: fatherRole admin can't equal role");
      require(role != fatherRole, "SC106: role == fatherRole");

      fatherRoleMap[role] = fatherRole;
    }

    function setRoleAdmin(bytes32 role, bytes32 adminRole) public isSuperAdmin checkNotAdminRole(role){
      require(fatherRoleMap[adminRole] != role, "SC105: adminRole's admin can't equal role");
      require(role != adminRole, "SC106: role == adminRole");

      _setRoleAdmin(role, adminRole);
    }

    function grantRole(bytes32 role, address account) public override isRoleAdmin(role){
      super._grantRole(role, account);
    }

    function getRawValue(bytes32 key) public view override returns(bytes32 typeID, bytes memory data){
        return _configSet._get(key);
    }

    function getKey(string memory keyStr) public pure override returns(bytes32 key){
        return LEnumerableMetadata._getKeyID(keyStr);
    }

    function getAllkeys(string memory startKey, uint256 pageSize) public view returns (string[] memory keys) {
        return _configSet._getAllKeys(startKey, pageSize);
    }

    function setKVs(bytes[] memory mds) public isAdmin {
        _configSet._setBytesSlice(mds);
        for (uint256 i = 0; i < mds.length; i++) {
            (string memory keyStr, bytes32 typeID, bytes memory data) = abi.decode(mds[i], (string, bytes32, bytes));
            emit SetKVEvent(getKey(keyStr), keyStr, typeID, data);
        }
    }

}


// File: contracts/config/IGlobalConfig.sol



// SPDX-License-Identifier: MIT
pragma solidity >=0.8.0 <0.9.0;

import "./IConfig.sol";

interface IGlobalConfig is IConfig {

  event SetKVEvent(bytes32 indexed key, string keyStr, bytes32 typeID, bytes data);

  function getKey(string memory keyStr) external pure returns(bytes32 key);
  
  function setKVs(bytes[] memory mds) external;

}

// File: contracts/library/Registry.sol



// SPDX-License-Identifier: MIT
pragma solidity >=0.8.0 <0.9.0;

library Registry {

  /***************************** ROLE NAME CONSTANT VARIABLES  ***********************************/

  // SUPER_ADMIN_ROLE
  bytes32 internal constant SUPER_ADMIN_ROLE = 0x0000000000000000000000000000000000000000000000000000000000000000;

  // keccak256("ADMIN_ROLE");
  bytes32 internal constant ADMIN_ROLE = 0xa49807205ce4d355092ef5a8a18f56e8913cf4a201fbe287825b095693c21775;

  // keccak256("only.on.chain");
  bytes32 internal constant ONLY_ON_CHAIN = 0x0b7322ebc56f27a124c063802688bbc3635a2ce3f24abda3e4adaec5fce31196;
}

// File: contracts/library/LEnumerableMetadata.sol



//SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

import "./LChainLink.sol";

library LEnumerableMetadata {
    using LChainLink for LChainLink.Link;

    // chain.
    // "map key" is keccak256("key")
    struct Metadata {
        string key;
        bytes32 typeID;
        bytes data;
        LChainLink.Link link;
    }

    struct MetadataSet {
        mapping(bytes32 => Metadata) kvData;
        uint256 size;
    }

    function _getKeyID(string memory key) internal pure returns (bytes32) {
        require(bytes(key).length != 0, "key is empty");

        bytes32 id = keccak256(abi.encodePacked(key));

        require(!LChainLink._isEmpty(id), "id error");
        return id;
    }

    function _init(MetadataSet storage m) internal {
        m.kvData[LChainLink.SENTINEL_ID].link._init();
    }

    function _checkKvDatas(bytes[] memory kvDatas) internal pure {
        for (uint256 i = 0; i < kvDatas.length; i++) {
            _parseKvData(kvDatas[i]);
        }
    }

    function _parseKvData(bytes memory data)
        internal
        pure
        returns (
            string memory,
            bytes32,
            bytes memory
        )
    {
        return abi.decode(data, (string, bytes32, bytes));
    }

    function _encodeKvData(
        string memory key,
        bytes32 typeID,
        bytes memory data,
        string memory desc
    ) internal pure returns (bytes memory) {
        return abi.encode(key, typeID, data, desc);
    }

    function _setBytesSlice(MetadataSet storage m, bytes[] memory mds)
        internal
    {
        for (uint256 i = 0; i < mds.length; i++) {
            (
                string memory key,
                bytes32 typeID,
                bytes memory data
            ) = _parseKvData(mds[i]);

            if (data.length == 0) {
                _del(m, key);
            } else {
                _addOrChange(m, key, typeID, data);
            }
        }
    }

    function _addOrChange(
        MetadataSet storage m,
        string memory key,
        bytes32 typeID,
        bytes memory data
    ) internal {
        bytes32 keyID = _getKeyID(key);

        Metadata storage md = m.kvData[keyID];

        if (!md.link._isEmpty()) {
            // change data
            if (md.typeID != typeID) {
                md.typeID = typeID;
            }

            md.data = data;

            return;
        }

        // add data
        md.key = key;
        md.typeID = typeID;
        md.data = data;

        Metadata storage sentinel = m.kvData[LChainLink.SENTINEL_ID];
        md.link._addItemLink(
            sentinel.link,
            m.kvData[sentinel.link.next].link,
            keyID
        );

        m.size++;
    }

    function _del(MetadataSet storage m, string memory key) internal {
        bytes32 keyID = _getKeyID(key);
        Metadata storage md = m.kvData[keyID];

        if (md.link._isEmpty()) {
            return;
        }

        md.link._delItemLink(
            m.kvData[md.link.pre].link,
            m.kvData[md.link.next].link
        );

        delete m.kvData[keyID];
        m.size--;
    }

    function _getByKey(MetadataSet storage m, string memory key)
        internal
        view
        returns (bytes32 typeID, bytes memory data)
    {
        return _get(m, _getKeyID(key));
    }

    function _get(MetadataSet storage m, bytes32 keyID)
        internal
        view
        returns (bytes32 typeID, bytes memory data)
    {
        Metadata storage md = m.kvData[keyID];

        return (md.typeID, md.data);
    }

    function _getAllKeys(
        MetadataSet storage m,
        string memory startKey,
        uint256 pageSize
    ) internal view returns (string[] memory keys) {
        keys = new string[](pageSize);
        uint256 idx = 0;

        bytes32 nowKeyID;
        if (bytes(startKey).length == 0) {
            nowKeyID = LChainLink.SENTINEL_ID;
        } else {
            nowKeyID = _getKeyID(startKey);
        }

        nowKeyID = m.kvData[nowKeyID].link._getNextID();
        while (idx < pageSize && !LChainLink._isEmpty(nowKeyID)) {
            Metadata storage md = m.kvData[nowKeyID];

            keys[idx] = md.key;

            nowKeyID = md.link._getNextID();
            idx++;
        }

        assembly {
            mstore(keys, idx)
        }

        return keys;
    }

    // return encode(string key, bytes32 typeID, bytes value)[]
    function _getAll(
        MetadataSet storage m,
        string memory startKey,
        uint256 pageSize
    ) internal view returns (bytes[] memory datas) {
        datas = new bytes[](pageSize);
        uint256 idx = 0;

        bytes32 nowKeyID;
        if (bytes(startKey).length == 0) {
            nowKeyID = LChainLink.SENTINEL_ID;
        } else {
            nowKeyID = _getKeyID(startKey);
        }

        nowKeyID = m.kvData[nowKeyID].link._getNextID();
        while (idx < pageSize && !LChainLink._isEmpty(nowKeyID)) {
            Metadata storage md = m.kvData[nowKeyID];

            datas[idx] = abi.encode(md.key, md.typeID, md.data);

            nowKeyID = md.link._getNextID();
            idx++;
        }

        assembly {
            mstore(datas, idx)
        }

        return datas;
    }
}
// File: @openzeppelin/contracts/access/AccessControlEnumerable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.5.0) (access/AccessControlEnumerable.sol)

pragma solidity ^0.8.0;

import "./IAccessControlEnumerable.sol";
import "./AccessControl.sol";
import "../utils/structs/EnumerableSet.sol";

/**
 * @dev Extension of {AccessControl} that allows enumerating the members of each role.
 */
abstract contract AccessControlEnumerable is IAccessControlEnumerable, AccessControl {
    using EnumerableSet for EnumerableSet.AddressSet;

    mapping(bytes32 => EnumerableSet.AddressSet) private _roleMembers;

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IAccessControlEnumerable).interfaceId || super.supportsInterface(interfaceId);
    }

    /**
     * @dev Returns one of the accounts that have `role`. `index` must be a
     * value between 0 and {getRoleMemberCount}, non-inclusive.
     *
     * Role bearers are not sorted in any particular way, and their ordering may
     * change at any point.
     *
     * WARNING: When using {getRoleMember} and {getRoleMemberCount}, make sure
     * you perform all queries on the same block. See the following
     * https://forum.openzeppelin.com/t/iterating-over-elements-on-enumerableset-in-openzeppelin-contracts/2296[forum post]
     * for more information.
     */
    function getRoleMember(bytes32 role, uint256 index) public view virtual override returns (address) {
        return _roleMembers[role].at(index);
    }

    /**
     * @dev Returns the number of accounts that have `role`. Can be used
     * together with {getRoleMember} to enumerate all bearers of a role.
     */
    function getRoleMemberCount(bytes32 role) public view virtual override returns (uint256) {
        return _roleMembers[role].length();
    }

    /**
     * @dev Overload {_grantRole} to track enumerable memberships
     */
    function _grantRole(bytes32 role, address account) internal virtual override {
        super._grantRole(role, account);
        _roleMembers[role].add(account);
    }

    /**
     * @dev Overload {_revokeRole} to track enumerable memberships
     */
    function _revokeRole(bytes32 role, address account) internal virtual override {
        super._revokeRole(role, account);
        _roleMembers[role].remove(account);
    }
}

// File: contracts/config/IConfig.sol



// SPDX-License-Identifier: MIT
pragma solidity >=0.8.0 <0.9.0;

interface IConfig {
  function version() external pure returns (uint256 v);

  function getRawValue(bytes32 key) external view returns(bytes32 typeID, bytes memory data);

  function hasRole(bytes32 role, address account) external view returns(bool has);

  function supportsInterface(bytes4 interfaceId) external view returns (bool);
  
}

// File: contracts/library/LChainLink.sol



//SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

library LChainLink {
    struct Link {
        bytes32 id;
        bytes32 next;
        bytes32 pre;
    }

    bytes32 internal constant SENTINEL_ID =
        bytes32(
            0x0000000000000000000000000000000000000000000000000000000000000001
        );

    address internal constant SENTINEL_ADDR = address(0x1);

    //////////////////// for bytes32

    function _isEmpty(bytes32 id) internal pure returns (bool) {
        if (id == bytes32(0x0) || id == SENTINEL_ID) {
            return true;
        }

        return false;
    }

    function _isEmpty(Link storage link) internal view returns (bool) {
        if (
            link.id == bytes32(0x0) ||
            (link.next == link.pre && link.next == bytes32(0x0))
        ) {
            return true;
        }

        return false;
    }

    function _init(Link storage link) internal {
        if (!_isEmpty(link)) {
            return;
        }

        link.id = SENTINEL_ID;
        link.next = SENTINEL_ID;
        link.pre = SENTINEL_ID;
    }

    function _addItemLink(
        Link storage item,
        Link storage sentinel,
        Link storage next,
        bytes32 itemID
    ) internal {
        require(
            sentinel.next == next.id && next.pre == SENTINEL_ID,
            "add link error"
        );

        item.id = itemID;
        item.next = sentinel.next;
        item.pre = SENTINEL_ID;

        next.pre = itemID;
        sentinel.next = itemID;
    }

    function _delItemLink(
        Link storage item,
        Link storage pre,
        Link storage next
    ) internal {
        require(pre.next == item.id && next.pre == item.id, "del link error");
        item.id = bytes32(0x0);
        item.next = bytes32(0x0);
        item.pre = bytes32(0x0);

        pre.next = next.id;
        next.pre = pre.id;
    }

    function _getNextID(Link storage item) internal view returns (bytes32) {
        // require(item.next != bytes32(0x0), "get link next error");
        return item.next;
    }

    function _getPreID(Link storage item) internal view returns (bytes32) {
        // require(item.pre != bytes32(0x0), "get link pre error");
        return item.pre;
    }

    //////////////////// for address
    function _isEmpty(address addr) internal pure returns (bool) {
        return _isEmpty(bytes32(uint256(uint160(addr))));
    }

    function _addItemLink(
        Link storage item,
        Link storage sentinel,
        Link storage next,
        address itemAddr
    ) internal {
        _addItemLink(item, sentinel, next, bytes32(uint256(uint160(itemAddr))));
    }

    function _getNextAddr(Link storage item) internal view returns (address) {
        return address(uint160(uint256(_getNextID(item))));
    }

    function _getPreAddr(Link storage item) internal view returns (address) {
        return address(uint160(uint256(_getPreID(item))));
    }
}
// File: @openzeppelin/contracts/access/IAccessControlEnumerable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (access/IAccessControlEnumerable.sol)

pragma solidity ^0.8.0;

import "./IAccessControl.sol";

/**
 * @dev External interface of AccessControlEnumerable declared to support ERC165 detection.
 */
interface IAccessControlEnumerable is IAccessControl {
    /**
     * @dev Returns one of the accounts that have `role`. `index` must be a
     * value between 0 and {getRoleMemberCount}, non-inclusive.
     *
     * Role bearers are not sorted in any particular way, and their ordering may
     * change at any point.
     *
     * WARNING: When using {getRoleMember} and {getRoleMemberCount}, make sure
     * you perform all queries on the same block. See the following
     * https://forum.openzeppelin.com/t/iterating-over-elements-on-enumerableset-in-openzeppelin-contracts/2296[forum post]
     * for more information.
     */
    function getRoleMember(bytes32 role, uint256 index) external view returns (address);

    /**
     * @dev Returns the number of accounts that have `role`. Can be used
     * together with {getRoleMember} to enumerate all bearers of a role.
     */
    function getRoleMemberCount(bytes32 role) external view returns (uint256);
}

// File: @openzeppelin/contracts/access/AccessControl.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/AccessControl.sol)

pragma solidity ^0.8.0;

import "./IAccessControl.sol";
import "../utils/Context.sol";
import "../utils/Strings.sol";
import "../utils/introspection/ERC165.sol";

/**
 * @dev Contract module that allows children to implement role-based access
 * control mechanisms. This is a lightweight version that doesn't allow enumerating role
 * members except through off-chain means by accessing the contract event logs. Some
 * applications may benefit from on-chain enumerability, for those cases see
 * {AccessControlEnumerable}.
 *
 * Roles are referred to by their `bytes32` identifier. These should be exposed
 * in the external API and be unique. The best way to achieve this is by
 * using `public constant` hash digests:
 *
 * ```
 * bytes32 public constant MY_ROLE = keccak256("MY_ROLE");
 * ```
 *
 * Roles can be used to represent a set of permissions. To restrict access to a
 * function call, use {hasRole}:
 *
 * ```
 * function foo() public {
 *     require(hasRole(MY_ROLE, msg.sender));
 *     ...
 * }
 * ```
 *
 * Roles can be granted and revoked dynamically via the {grantRole} and
 * {revokeRole} functions. Each role has an associated admin role, and only
 * accounts that have a role's admin role can call {grantRole} and {revokeRole}.
 *
 * By default, the admin role for all roles is `DEFAULT_ADMIN_ROLE`, which means
 * that only accounts with this role will be able to grant or revoke other
 * roles. More complex role relationships can be created by using
 * {_setRoleAdmin}.
 *
 * WARNING: The `DEFAULT_ADMIN_ROLE` is also its own admin: it has permission to
 * grant and revoke this role. Extra precautions should be taken to secure
 * accounts that have been granted it.
 */
abstract contract AccessControl is Context, IAccessControl, ERC165 {
    struct RoleData {
        mapping(address => bool) members;
        bytes32 adminRole;
    }

    mapping(bytes32 => RoleData) private _roles;

    bytes32 public constant DEFAULT_ADMIN_ROLE = 0x00;

    /**
     * @dev Modifier that checks that an account has a specific role. Reverts
     * with a standardized message including the required role.
     *
     * The format of the revert reason is given by the following regular expression:
     *
     *  /^AccessControl: account (0x[0-9a-f]{40}) is missing role (0x[0-9a-f]{64})$/
     *
     * _Available since v4.1._
     */
    modifier onlyRole(bytes32 role) {
        _checkRole(role);
        _;
    }

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IAccessControl).interfaceId || super.supportsInterface(interfaceId);
    }

    /**
     * @dev Returns `true` if `account` has been granted `role`.
     */
    function hasRole(bytes32 role, address account) public view virtual override returns (bool) {
        return _roles[role].members[account];
    }

    /**
     * @dev Revert with a standard message if `_msgSender()` is missing `role`.
     * Overriding this function changes the behavior of the {onlyRole} modifier.
     *
     * Format of the revert message is described in {_checkRole}.
     *
     * _Available since v4.6._
     */
    function _checkRole(bytes32 role) internal view virtual {
        _checkRole(role, _msgSender());
    }

    /**
     * @dev Revert with a standard message if `account` is missing `role`.
     *
     * The format of the revert reason is given by the following regular expression:
     *
     *  /^AccessControl: account (0x[0-9a-f]{40}) is missing role (0x[0-9a-f]{64})$/
     */
    function _checkRole(bytes32 role, address account) internal view virtual {
        if (!hasRole(role, account)) {
            revert(
                string(
                    abi.encodePacked(
                        "AccessControl: account ",
                        Strings.toHexString(uint160(account), 20),
                        " is missing role ",
                        Strings.toHexString(uint256(role), 32)
                    )
                )
            );
        }
    }

    /**
     * @dev Returns the admin role that controls `role`. See {grantRole} and
     * {revokeRole}.
     *
     * To change a role's admin, use {_setRoleAdmin}.
     */
    function getRoleAdmin(bytes32 role) public view virtual override returns (bytes32) {
        return _roles[role].adminRole;
    }

    /**
     * @dev Grants `role` to `account`.
     *
     * If `account` had not been already granted `role`, emits a {RoleGranted}
     * event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     *
     * May emit a {RoleGranted} event.
     */
    function grantRole(bytes32 role, address account) public virtual override onlyRole(getRoleAdmin(role)) {
        _grantRole(role, account);
    }

    /**
     * @dev Revokes `role` from `account`.
     *
     * If `account` had been granted `role`, emits a {RoleRevoked} event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     *
     * May emit a {RoleRevoked} event.
     */
    function revokeRole(bytes32 role, address account) public virtual override onlyRole(getRoleAdmin(role)) {
        _revokeRole(role, account);
    }

    /**
     * @dev Revokes `role` from the calling account.
     *
     * Roles are often managed via {grantRole} and {revokeRole}: this function's
     * purpose is to provide a mechanism for accounts to lose their privileges
     * if they are compromised (such as when a trusted device is misplaced).
     *
     * If the calling account had been revoked `role`, emits a {RoleRevoked}
     * event.
     *
     * Requirements:
     *
     * - the caller must be `account`.
     *
     * May emit a {RoleRevoked} event.
     */
    function renounceRole(bytes32 role, address account) public virtual override {
        require(account == _msgSender(), "AccessControl: can only renounce roles for self");

        _revokeRole(role, account);
    }

    /**
     * @dev Grants `role` to `account`.
     *
     * If `account` had not been already granted `role`, emits a {RoleGranted}
     * event. Note that unlike {grantRole}, this function doesn't perform any
     * checks on the calling account.
     *
     * May emit a {RoleGranted} event.
     *
     * [WARNING]
     * ====
     * This function should only be called from the constructor when setting
     * up the initial roles for the system.
     *
     * Using this function in any other way is effectively circumventing the admin
     * system imposed by {AccessControl}.
     * ====
     *
     * NOTE: This function is deprecated in favor of {_grantRole}.
     */
    function _setupRole(bytes32 role, address account) internal virtual {
        _grantRole(role, account);
    }

    /**
     * @dev Sets `adminRole` as ``role``'s admin role.
     *
     * Emits a {RoleAdminChanged} event.
     */
    function _setRoleAdmin(bytes32 role, bytes32 adminRole) internal virtual {
        bytes32 previousAdminRole = getRoleAdmin(role);
        _roles[role].adminRole = adminRole;
        emit RoleAdminChanged(role, previousAdminRole, adminRole);
    }

    /**
     * @dev Grants `role` to `account`.
     *
     * Internal function without access restriction.
     *
     * May emit a {RoleGranted} event.
     */
    function _grantRole(bytes32 role, address account) internal virtual {
        if (!hasRole(role, account)) {
            _roles[role].members[account] = true;
            emit RoleGranted(role, account, _msgSender());
        }
    }

    /**
     * @dev Revokes `role` from `account`.
     *
     * Internal function without access restriction.
     *
     * May emit a {RoleRevoked} event.
     */
    function _revokeRole(bytes32 role, address account) internal virtual {
        if (hasRole(role, account)) {
            _roles[role].members[account] = false;
            emit RoleRevoked(role, account, _msgSender());
        }
    }
}

// File: @openzeppelin/contracts/utils/structs/EnumerableSet.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (utils/structs/EnumerableSet.sol)

pragma solidity ^0.8.0;

/**
 * @dev Library for managing
 * https://en.wikipedia.org/wiki/Set_(abstract_data_type)[sets] of primitive
 * types.
 *
 * Sets have the following properties:
 *
 * - Elements are added, removed, and checked for existence in constant time
 * (O(1)).
 * - Elements are enumerated in O(n). No guarantees are made on the ordering.
 *
 * ```
 * contract Example {
 *     // Add the library methods
 *     using EnumerableSet for EnumerableSet.AddressSet;
 *
 *     // Declare a set state variable
 *     EnumerableSet.AddressSet private mySet;
 * }
 * ```
 *
 * As of v3.3.0, sets of type `bytes32` (`Bytes32Set`), `address` (`AddressSet`)
 * and `uint256` (`UintSet`) are supported.
 *
 * [WARNING]
 * ====
 *  Trying to delete such a structure from storage will likely result in data corruption, rendering the structure unusable.
 *  See https://github.com/ethereum/solidity/pull/11843[ethereum/solidity#11843] for more info.
 *
 *  In order to clean an EnumerableSet, you can either remove all elements one by one or create a fresh instance using an array of EnumerableSet.
 * ====
 */
library EnumerableSet {
    // To implement this library for multiple types with as little code
    // repetition as possible, we write it in terms of a generic Set type with
    // bytes32 values.
    // The Set implementation uses private functions, and user-facing
    // implementations (such as AddressSet) are just wrappers around the
    // underlying Set.
    // This means that we can only create new EnumerableSets for types that fit
    // in bytes32.

    struct Set {
        // Storage of set values
        bytes32[] _values;
        // Position of the value in the `values` array, plus 1 because index 0
        // means a value is not in the set.
        mapping(bytes32 => uint256) _indexes;
    }

    /**
     * @dev Add a value to a set. O(1).
     *
     * Returns true if the value was added to the set, that is if it was not
     * already present.
     */
    function _add(Set storage set, bytes32 value) private returns (bool) {
        if (!_contains(set, value)) {
            set._values.push(value);
            // The value is stored at length-1, but we add 1 to all indexes
            // and use 0 as a sentinel value
            set._indexes[value] = set._values.length;
            return true;
        } else {
            return false;
        }
    }

    /**
     * @dev Removes a value from a set. O(1).
     *
     * Returns true if the value was removed from the set, that is if it was
     * present.
     */
    function _remove(Set storage set, bytes32 value) private returns (bool) {
        // We read and store the value's index to prevent multiple reads from the same storage slot
        uint256 valueIndex = set._indexes[value];

        if (valueIndex != 0) {
            // Equivalent to contains(set, value)
            // To delete an element from the _values array in O(1), we swap the element to delete with the last one in
            // the array, and then remove the last element (sometimes called as 'swap and pop').
            // This modifies the order of the array, as noted in {at}.

            uint256 toDeleteIndex = valueIndex - 1;
            uint256 lastIndex = set._values.length - 1;

            if (lastIndex != toDeleteIndex) {
                bytes32 lastValue = set._values[lastIndex];

                // Move the last value to the index where the value to delete is
                set._values[toDeleteIndex] = lastValue;
                // Update the index for the moved value
                set._indexes[lastValue] = valueIndex; // Replace lastValue's index to valueIndex
            }

            // Delete the slot where the moved value was stored
            set._values.pop();

            // Delete the index for the deleted slot
            delete set._indexes[value];

            return true;
        } else {
            return false;
        }
    }

    /**
     * @dev Returns true if the value is in the set. O(1).
     */
    function _contains(Set storage set, bytes32 value) private view returns (bool) {
        return set._indexes[value] != 0;
    }

    /**
     * @dev Returns the number of values on the set. O(1).
     */
    function _length(Set storage set) private view returns (uint256) {
        return set._values.length;
    }

    /**
     * @dev Returns the value stored at position `index` in the set. O(1).
     *
     * Note that there are no guarantees on the ordering of values inside the
     * array, and it may change when more values are added or removed.
     *
     * Requirements:
     *
     * - `index` must be strictly less than {length}.
     */
    function _at(Set storage set, uint256 index) private view returns (bytes32) {
        return set._values[index];
    }

    /**
     * @dev Return the entire set in an array
     *
     * WARNING: This operation will copy the entire storage to memory, which can be quite expensive. This is designed
     * to mostly be used by view accessors that are queried without any gas fees. Developers should keep in mind that
     * this function has an unbounded cost, and using it as part of a state-changing function may render the function
     * uncallable if the set grows to a point where copying to memory consumes too much gas to fit in a block.
     */
    function _values(Set storage set) private view returns (bytes32[] memory) {
        return set._values;
    }

    // Bytes32Set

    struct Bytes32Set {
        Set _inner;
    }

    /**
     * @dev Add a value to a set. O(1).
     *
     * Returns true if the value was added to the set, that is if it was not
     * already present.
     */
    function add(Bytes32Set storage set, bytes32 value) internal returns (bool) {
        return _add(set._inner, value);
    }

    /**
     * @dev Removes a value from a set. O(1).
     *
     * Returns true if the value was removed from the set, that is if it was
     * present.
     */
    function remove(Bytes32Set storage set, bytes32 value) internal returns (bool) {
        return _remove(set._inner, value);
    }

    /**
     * @dev Returns true if the value is in the set. O(1).
     */
    function contains(Bytes32Set storage set, bytes32 value) internal view returns (bool) {
        return _contains(set._inner, value);
    }

    /**
     * @dev Returns the number of values in the set. O(1).
     */
    function length(Bytes32Set storage set) internal view returns (uint256) {
        return _length(set._inner);
    }

    /**
     * @dev Returns the value stored at position `index` in the set. O(1).
     *
     * Note that there are no guarantees on the ordering of values inside the
     * array, and it may change when more values are added or removed.
     *
     * Requirements:
     *
     * - `index` must be strictly less than {length}.
     */
    function at(Bytes32Set storage set, uint256 index) internal view returns (bytes32) {
        return _at(set._inner, index);
    }

    /**
     * @dev Return the entire set in an array
     *
     * WARNING: This operation will copy the entire storage to memory, which can be quite expensive. This is designed
     * to mostly be used by view accessors that are queried without any gas fees. Developers should keep in mind that
     * this function has an unbounded cost, and using it as part of a state-changing function may render the function
     * uncallable if the set grows to a point where copying to memory consumes too much gas to fit in a block.
     */
    function values(Bytes32Set storage set) internal view returns (bytes32[] memory) {
        return _values(set._inner);
    }

    // AddressSet

    struct AddressSet {
        Set _inner;
    }

    /**
     * @dev Add a value to a set. O(1).
     *
     * Returns true if the value was added to the set, that is if it was not
     * already present.
     */
    function add(AddressSet storage set, address value) internal returns (bool) {
        return _add(set._inner, bytes32(uint256(uint160(value))));
    }

    /**
     * @dev Removes a value from a set. O(1).
     *
     * Returns true if the value was removed from the set, that is if it was
     * present.
     */
    function remove(AddressSet storage set, address value) internal returns (bool) {
        return _remove(set._inner, bytes32(uint256(uint160(value))));
    }

    /**
     * @dev Returns true if the value is in the set. O(1).
     */
    function contains(AddressSet storage set, address value) internal view returns (bool) {
        return _contains(set._inner, bytes32(uint256(uint160(value))));
    }

    /**
     * @dev Returns the number of values in the set. O(1).
     */
    function length(AddressSet storage set) internal view returns (uint256) {
        return _length(set._inner);
    }

    /**
     * @dev Returns the value stored at position `index` in the set. O(1).
     *
     * Note that there are no guarantees on the ordering of values inside the
     * array, and it may change when more values are added or removed.
     *
     * Requirements:
     *
     * - `index` must be strictly less than {length}.
     */
    function at(AddressSet storage set, uint256 index) internal view returns (address) {
        return address(uint160(uint256(_at(set._inner, index))));
    }

    /**
     * @dev Return the entire set in an array
     *
     * WARNING: This operation will copy the entire storage to memory, which can be quite expensive. This is designed
     * to mostly be used by view accessors that are queried without any gas fees. Developers should keep in mind that
     * this function has an unbounded cost, and using it as part of a state-changing function may render the function
     * uncallable if the set grows to a point where copying to memory consumes too much gas to fit in a block.
     */
    function values(AddressSet storage set) internal view returns (address[] memory) {
        bytes32[] memory store = _values(set._inner);
        address[] memory result;

        /// @solidity memory-safe-assembly
        assembly {
            result := store
        }

        return result;
    }

    // UintSet

    struct UintSet {
        Set _inner;
    }

    /**
     * @dev Add a value to a set. O(1).
     *
     * Returns true if the value was added to the set, that is if it was not
     * already present.
     */
    function add(UintSet storage set, uint256 value) internal returns (bool) {
        return _add(set._inner, bytes32(value));
    }

    /**
     * @dev Removes a value from a set. O(1).
     *
     * Returns true if the value was removed from the set, that is if it was
     * present.
     */
    function remove(UintSet storage set, uint256 value) internal returns (bool) {
        return _remove(set._inner, bytes32(value));
    }

    /**
     * @dev Returns true if the value is in the set. O(1).
     */
    function contains(UintSet storage set, uint256 value) internal view returns (bool) {
        return _contains(set._inner, bytes32(value));
    }

    /**
     * @dev Returns the number of values on the set. O(1).
     */
    function length(UintSet storage set) internal view returns (uint256) {
        return _length(set._inner);
    }

    /**
     * @dev Returns the value stored at position `index` in the set. O(1).
     *
     * Note that there are no guarantees on the ordering of values inside the
     * array, and it may change when more values are added or removed.
     *
     * Requirements:
     *
     * - `index` must be strictly less than {length}.
     */
    function at(UintSet storage set, uint256 index) internal view returns (uint256) {
        return uint256(_at(set._inner, index));
    }

    /**
     * @dev Return the entire set in an array
     *
     * WARNING: This operation will copy the entire storage to memory, which can be quite expensive. This is designed
     * to mostly be used by view accessors that are queried without any gas fees. Developers should keep in mind that
     * this function has an unbounded cost, and using it as part of a state-changing function may render the function
     * uncallable if the set grows to a point where copying to memory consumes too much gas to fit in a block.
     */
    function values(UintSet storage set) internal view returns (uint256[] memory) {
        bytes32[] memory store = _values(set._inner);
        uint256[] memory result;

        /// @solidity memory-safe-assembly
        assembly {
            result := store
        }

        return result;
    }
}

// File: @openzeppelin/contracts/access/IAccessControl.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (access/IAccessControl.sol)

pragma solidity ^0.8.0;

/**
 * @dev External interface of AccessControl declared to support ERC165 detection.
 */
interface IAccessControl {
    /**
     * @dev Emitted when `newAdminRole` is set as ``role``'s admin role, replacing `previousAdminRole`
     *
     * `DEFAULT_ADMIN_ROLE` is the starting admin for all roles, despite
     * {RoleAdminChanged} not being emitted signaling this.
     *
     * _Available since v3.1._
     */
    event RoleAdminChanged(bytes32 indexed role, bytes32 indexed previousAdminRole, bytes32 indexed newAdminRole);

    /**
     * @dev Emitted when `account` is granted `role`.
     *
     * `sender` is the account that originated the contract call, an admin role
     * bearer except when using {AccessControl-_setupRole}.
     */
    event RoleGranted(bytes32 indexed role, address indexed account, address indexed sender);

    /**
     * @dev Emitted when `account` is revoked `role`.
     *
     * `sender` is the account that originated the contract call:
     *   - if using `revokeRole`, it is the admin role bearer
     *   - if using `renounceRole`, it is the role bearer (i.e. `account`)
     */
    event RoleRevoked(bytes32 indexed role, address indexed account, address indexed sender);

    /**
     * @dev Returns `true` if `account` has been granted `role`.
     */
    function hasRole(bytes32 role, address account) external view returns (bool);

    /**
     * @dev Returns the admin role that controls `role`. See {grantRole} and
     * {revokeRole}.
     *
     * To change a role's admin, use {AccessControl-_setRoleAdmin}.
     */
    function getRoleAdmin(bytes32 role) external view returns (bytes32);

    /**
     * @dev Grants `role` to `account`.
     *
     * If `account` had not been already granted `role`, emits a {RoleGranted}
     * event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     */
    function grantRole(bytes32 role, address account) external;

    /**
     * @dev Revokes `role` from `account`.
     *
     * If `account` had been granted `role`, emits a {RoleRevoked} event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     */
    function revokeRole(bytes32 role, address account) external;

    /**
     * @dev Revokes `role` from the calling account.
     *
     * Roles are often managed via {grantRole} and {revokeRole}: this function's
     * purpose is to provide a mechanism for accounts to lose their privileges
     * if they are compromised (such as when a trusted device is misplaced).
     *
     * If the calling account had been granted `role`, emits a {RoleRevoked}
     * event.
     *
     * Requirements:
     *
     * - the caller must be `account`.
     */
    function renounceRole(bytes32 role, address account) external;
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/utils/Strings.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (utils/Strings.sol)

pragma solidity ^0.8.0;

/**
 * @dev String operations.
 */
library Strings {
    bytes16 private constant _HEX_SYMBOLS = "0123456789abcdef";
    uint8 private constant _ADDRESS_LENGTH = 20;

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        // Inspired by OraclizeAPI's implementation - MIT licence
        // https://github.com/oraclize/ethereum-api/blob/b42146b063c7d6ee1358846c198246239e9360e8/oraclizeAPI_0.4.25.sol

        if (value == 0) {
            return "0";
        }
        uint256 temp = value;
        uint256 digits;
        while (temp != 0) {
            digits++;
            temp /= 10;
        }
        bytes memory buffer = new bytes(digits);
        while (value != 0) {
            digits -= 1;
            buffer[digits] = bytes1(uint8(48 + uint256(value % 10)));
            value /= 10;
        }
        return string(buffer);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        if (value == 0) {
            return "0x00";
        }
        uint256 temp = value;
        uint256 length = 0;
        while (temp != 0) {
            length++;
            temp >>= 8;
        }
        return toHexString(value, length);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = _HEX_SYMBOLS[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }

    /**
     * @dev Converts an `address` with fixed length of 20 bytes to its not checksummed ASCII `string` hexadecimal representation.
     */
    function toHexString(address addr) internal pure returns (string memory) {
        return toHexString(uint256(uint160(addr)), _ADDRESS_LENGTH);
    }
}

// File: @openzeppelin/contracts/utils/introspection/ERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/ERC165.sol)

pragma solidity ^0.8.0;

import "./IERC165.sol";

/**
 * @dev Implementation of the {IERC165} interface.
 *
 * Contracts that want to implement ERC165 should inherit from this contract and override {supportsInterface} to check
 * for the additional interface id that will be supported. For example:
 *
 * ```solidity
 * function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
 *     return interfaceId == type(MyInterface).interfaceId || super.supportsInterface(interfaceId);
 * }
 * ```
 *
 * Alternatively, {ERC165Storage} provides an easier to use but more expensive implementation.
 */
abstract contract ERC165 is IERC165 {
    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IERC165).interfaceId;
    }
}

// File: @openzeppelin/contracts/utils/introspection/IERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/IERC165.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}